txt: ## traits.md
  Traits
  
  Compile-time specification of behavior made easy
  
  In Raku, *traits* are compiler hooks attached to objects and classes that modify their default behavior, functionality or representation. As such compiler hooks, they are defined in compile time, although they can be used in runtime.
  
  Several traits are already defined as part of the language or the Rakudo compiler by using the `trait_mod` keyword. They are listed, and explained, next.
  
  # [`is` trait](#Traits "go to top of document")[§](#The_is_trait "direct link")
  
  ```raku
  proto trait_mod:<is>(Mu $, |) {*}
  ```
  
  `is` applies to any kind of scalar object, and can take any number of named or positional arguments. It is the most commonly used trait, and takes the following forms, depending on the type of the first argument.
  
  ## [`is` applied to classes.](#Traits "go to top of document")[§](#is_applied_to_classes. "direct link")
  
  The most common form, involving two classes, one that is being defined and the other existing, [defines parenthood](/routine/is). `A is B`, if both are classes, defines A as a subclass of B.
  
  [`is DEPRECATED`](/type/Attribute#trait_is_DEPRECATED) can be applied to classes, Attributes or Routines, marks them as deprecated and issues a message, if provided.
  
  Several instances of `is` are translated directly into attributes for the class they refer to: `rw`, `nativesize`, `ctype`, `unsigned`, `hidden`, `array_type`.
  
  The Uninstantiable representation trait is not so much related to the representation as related to what can be done with a specific class; it effectively prevents the creation of instances of the class in any possible way.
  
  ```raku
  constant @IMM = <Innie Minnie Moe>;
  
  class don't-instantiate is repr('Uninstantiable') {
      my $.counter;
  
      method imm () {
          return @IMM[ $.counter++ mod @IMM.elems ];
      }
  }
  say don't-instantiate.imm for ^10;
  
  ```
  
  Uninstantiable classes can still be used via their class variables and methods, as above. However, trying to instantiate them this way: `my $do-instantiate = don't-instantiate.new;` will yield the error `You cannot create an instance of this type (don't-instantiate)`.
  
  ## [`is repr` and native representations.](#Traits "go to top of document")[§](#is_repr_and_native_representations. "direct link")
  
  Since the `is` trait refers, in general, to the nature of the class or object they are applied to, they are used extensively in [native calls](/language/nativecall) to [specify the representation](/language/nativecall#Specifying_the_native_representation) of the data structures that are going to be handled by the native functions via the `is repr` suffix; at the same time, `is native` is used for the routines that are actually implemented via native functions. These are the representations that can be used:
  
  * CStruct corresponds to a `struct` in the C language. It is a composite data structure which includes different and heterogeneous lower-level data structures; see [this](/language/nativecall#Structs) for examples and further explanations.
  * CPPStruct, similarly, correspond to a `struct` in C++. However, this is Rakudo specific for the time being.
  * CPointer is a pointer in any of these languages. It is a dynamic data structure that must be instantiated before being used, can be [used](/language/nativecall#Basic_use_of_pointers) for classes whose methods are also native.
  * CUnion is going to use the same representation as an `union` in C; see [this](/language/nativecall#CUnions) for an example.
  
  On the other hand, P6opaque is the default representation used for all objects in Raku.
  
  ```raku
  class Thar {};
  say Thar.REPR;    # OUTPUT: «P6opaque␤»
  ```
  
  The [metaobject protocol](/language/mop) uses it by default for every object and class unless specified otherwise; for that reason, it is in general not necessary unless you are effectively working with that interface.
  
  ## [`is` on routines](#Traits "go to top of document")[§](#is_on_routines "direct link")
  
  The `is` trait can be used on the definition of methods and routines to establish [precedence](/language/functions#Precedence) and [associativity](/language/functions#Associativity). They act as a [sub defined using `trait_mod`](/type/Sub#Traits) which take as argument the types and names of the traits that are going to be added. In the case of subroutines, traits would be a way of adding functionality which cuts across class and role hierarchies, or can even be used to add behaviors to independently defined routines.
  
  ## [`is implementation-detail`](#Traits "go to top of document")[§](#is_implementation-detail_trait "direct link")
  
  Available as of the 2020.05 release of the Rakudo compiler.
  
  This trait is used by Raku language implementations and module authors to mark particular routines (including methods) as not meant to be a part of public API. While such routines can be found when looked up directly, they will not appear in results of introspection:
  
  ```raku
  my &do-not-use-routine = CORE::<&DYNAMIC>;
  say CORE::.keys.grep(* eq '&DYNAMIC'); # OUTPUT: «()␤»
  ```
  
  Such routines are not meant for use by users and their behavior and availability can be changed anytime.
  
  As of the 2021.02 release of the Rakudo compiler, it is also possible to apply the `is implementation-detail` method on classes and roles.
  
  ### [method is-implementation-detail](#Traits "go to top of document")[§](#method_is-implementation-detail "direct link")
  
  ```raku
  method is-implementation-detail(--> True)
  ```
  
  Applying this trait makes the `is-implementation-detail` method called on [`Code`](/type/Code) to return `True`, thus giving a hint to the user not to use it if they are not willing to maintain this code in case of changes for years to come:
  
  ```raku
  my &fail-routine = &fail;
  unless &fail-routine.is-implementation-detail {
      say "&fail is not an implementation detail, can expect backward compatibility";
  }
  
  sub PRIVATE-CALCULATION is implementation-detail { #`(Not safe to rely on this) }
  if &PRIVATE-CALCULATION.is-implementation-detail {
      say "You better not to rely on &PRIVATE-CALCULATION unless you really know what you are doing";
  }
  ```
  